

Low Power IoT Labs:

Building complete IoT Architectures with Terminals, Gateways, and Servers

In the following labs – **Lab1** and **Lab2** we are going to study and experiment with complete IoT architectures including terminal nodes (T) long range gateways (G) and IoT servers. To support the experimentation we are using multiple IoT platforms from **SmartComputerLab**.

Table of Contents

Lab 1: Simple IoT Architectures with CubeCell	3
1.1 Introduction.....	3
1.2 Hardware Architecture - main board (CubeCell).....	4
1.2.1 Software installation with the use Arduino board manager.....	5
1.2.2 Software installation via Git.....	6
Principal link for software resources:.....	6
1.3 Power consumption test with Power Profiler Kit II.....	6
1.3.1 PPK2 software installation.....	7
1.4 Testing CubeCell HTCC-AB01 main board.....	8
1.4.1 Board chip identifier.....	8
1.4.2 Integrated RGB LED.....	8
1.4.3 Integrated USR button.....	9
1.4.4 User flash memory.....	9
1.4.5 Sleep modes: timer and user interruption.....	10
1.4.5.1 Timer mode.....	10
1.4.5.2 User interrupt mode.....	11
1.4.6 System time.....	12
1.4.6.1 System time in active - high power mode.....	12
1.4.6.2 System time in deep_sleep - low power mode.....	12
1.4.7 WatchDog timer.....	13
To do.....	14
1.5 External components and low power operation.....	15
1.5.1 SSD1306 OLED.....	15
1.5.2 I2C device scan.....	16
1.5.3 SHT21 sensor module (I2C).....	17
1.5.3.1 Reading SHT21 sensor and display on OLED.....	18
To do:.....	18
1.5.3.2 Raw SHT21 driver.....	18
To do:.....	19
1.5.4 BH1750 – Ambient Light sensor.....	19
1.5.4.1 Raw BH1750 driver.....	20
To do.....	21
1.5.5 Air (CO2/TVOC) sensor CCS811 (SGP30).....	21
To do.....	22
1.5.6 “Time of Flight” distance sensors – VL53L0X and VL53L1X.....	22
1.5.6.1 VL53L03.....	22
1.5.6.2 VL53L1X.....	23
To do.....	24
1.5.7 SoftwareSerial – GPS - ATGM33.....	24
3.5.7.1 Simple – direct UART stream.....	24
1.5.7.2 Simple – direct UART stream and OLED display.....	25
1.5.7.3 UART stream decoded by TinyGPS++ library.....	25
To do.....	27
1.6 Low power operation period with lowPowerHandler().....	28
1.6.1 Low power operation with SHT21 sensor.....	28
To do.....	30
1.6.2 Direct GPS - UART stream and OLED display with low_power mode.....	31

Lab 2: LoRa modem - basics and low power operation.....	33
2.1 LoRa send with battery state value.....	33
2.1.1 Sending data packet (battery voltage).....	34
To do.....	35
2.2 Sending data packet with low_power management.....	36
To do.....	38
2.3 LoRa receiver - receiving the packets with battery state.....	39
To do.....	40
2.4 Sending LoRa packets with external sensor data.....	41
2.7 Terminal and the gateway codes.....	43
2.7.1 The code of terminal node.....	43
To do:.....	46
2.8 LoRa-WiFi Gateways for CubeCell terminals.....	47
2.8.1 LoRa receiver (+display) for CubeCell terminals.....	47
2.8.2 LoRa-WiFi gateway to ThingSpeak (+display) for CubeCell terminals.....	48
Assignment.....	50

In the following labs – **Lab3** and **Lab4** we are going to study and experiment with complete IoT architectures including terminal nodes (T) long range gateways (G) and IoT servers. To support the experimentation we are using multiple IoT platforms from **SmartComputerLab**.

These are

- CubeCell (ASR6501) platform with **ARM-Cortex M0** and **SX1262** LoRa modem
- ESP32 (Heltec) platform with WiFi – BT/BLE and **SX1276** LoRa modem
- Raspberry-Pi 3 and 4 boards with RAK2245 – **SX1301** multichannel LoRa modem

Both the CubeCell and ESP32 (Heltec) platforms operate with Nordic PPK2 – Power Profiler Kit II

During the first Lab3 we are introducing the CubeCell platform and first we are going to experiment with the board itself and different sensors (and display) attached to I2C bus. Then we will experiment with different communication functions related to LoRa modem including different radio parameters.

In this context we are going to use two boards in sender/receiver or client/server modes. The sender/client nodes operate in two phases :

- **high_power** (consumption) phase when sensors are activated and the radio transmission takes place.
- **low_power** phase when the sensors are de-activated (no power) and the MCU and LoRa modem enter into **deep-sleep state**

We will measure the value of the current flowing during these phases. To complete the measurements we are going to take into account different radio parameters for LoRa transmission such as signal bandwidth and spreading factor.

In the second part of this lab we will exploit the ESP32 (Heltec WiFi-LoRa V2) boards to build the LoRa-WiFi gateways and to communicate with the external IoT servers. In this context we will experiment with LoRaMQTT and LoRaTS protocols that are adapted directly to these applications.

These IoT gateways may be configured using different means, such as WiFi SmartConfig protocol, simple external EEPROM modules or micro-keyboards attached via UART interface.

During the second Lab4 we are going to study and experiment with **LoRaWAN** protocol and related gateways and devices. LoRaWAN is a standard protocol based on LoRa radio links and implemented by many telecom operators as well as by the open source communities.

We provide you with several LoRaWAN gateways combining Raspberry-Pi boards and SX1301 multichannel modems. In our case the RPI/SX1301 gateways will operate exclusively as packet forwarders.

We are going to exploit the LoRaWAN libraries running on CubeCell and ESP32 platforms to build some complete applications with several devices attached to each one.

You will be asked to analyze the power consumption of the terminal nodes (devices) and send the status of the battery to the [The Things Network](#) (TTN) network server. The TTN server is a kind of relay between the LoRaWAN gateways and the application servers such as ThingSpeak.

Lab 1 (Low Power IoT Labs)

Simple IoT Architectures with CubeCell (ARM+SX1262)

1.1 Introduction

This Lab (3-9 hours) is prepared to teach the low power technologies for IoT architectures and to experiment with CubeCell boards and LoRaWAN IoT DevKit from SmartComputerLab.

The main objective of the labs is to show how to implement a very low power consumption applications with the terminal nodes integrating small LiPo batteries and miniature solar panels. The proposed IoT configurations should operate over very long time periods (months-years).

Our IoT DevKit uses ARM-Cortex M0 and LoRa SX1262 modem combined into ASR6501 SoM (System on Module). The ASR6501 module is integrated into the CubeCell board.

Low power operation of the terminal nodes is based on several mechanism that allows us to put the MCU in **low_power** mode, to put the LoRa radio in **sleep mode**, and to **cut off** the current flow use by the sensors. Having all of these three conditions fulfilled we can downscale the overall current to 10-20 μA with 3.3V voltage.

This dormant state can be interrupted by the **timeout signal** starting the wake period with **high_power** consumption period. The **high_power** consumption period may be decomposed in several phases including sensing phase, transmission phase, reception phase with response data storage.

The energy consumption during the sensing phase depends on the characteristics of the sensor(s) including the sensing time and the current.

The energy consumption during the transmission phase depends on the transmission power and the duration of the transmission operation. These in turn depend on the size of the data packet and the LoRa radio parameters used for the transmission such as signal bandwidth, spreading factor and code-rate.

The duration of the **low_power** period may be fixed or provided dynamically by the gateway node through the time-out parameter sent in the response packet. This mechanism called **Target Wake Time (TWT)** is very efficient and also may be used to build the scheduled operation of a bigger number of terminal nodes. TWT based scheduling allows us to avoid the collisions between the transmitted packets.

The response also may carry the delta parameter informing the Terminal node about the variation of the sensor values to be taken into account for the transmission. For example we can imagine 3 different delta values for (1 for 0.1%, 10 for 1%, and 100 for 10%) the variation of the sensing values.

The following diagram shows the operational line of a terminal.




Figure 1.1 Low_power and high_power periods of operational cycle

1.2 Hardware Architecture - main board (CubeCell)

Heltec CubeCell board (HTCC-AB01) uses the ASR6501 module. The ASR6501 is SiP's (system-in-package) that combines a Cypress PSoC 4000 ARM Cortex-M0+ 32 bits 48MHz MCU (with 16kB SRAM and 128kB flash) together with a Semtech SX1262 LoRa transceiver in a single package.

The CubeCell products have an integrated LoRaWAN stack based on Semtech's LoRaMac-node.




Figure 1.2 ASR6501 architectural scheme with MCU and Lora modem connected via SPI bus




Figure 1.3 CellCube board and its pinout

for the implementation of the IoT terminals is mainly driven by the very low power consumption and the communication capabilities including LoRaWAN protocol.

The following are the power consumption characteristics:

- Supply current in Sleep mode: 10-20 μ A
- Supply current in Receiver mode: 10-20 mA
- Supply current in Transmitter mode: 80-100 mA

Imagine a terminal sending the data frames of 16 bytes once every 10 minutes ($10 \times 60 \times 1000$ ms). With the spreading factor of 8 and the bandwidth of 125kHz, the calculated airtime is: 123.4 ms (~ 125 ms)
It means that the average current consumption in mA is:

$$(125 \times 100 \text{ 000} + 10 \times 60 \times 1000) / 60 \times 1000 = (125 \times 100 + 10 \times 60) / 60 = (12500 + 60) / 60 = 210 \text{ } \mu\text{A} = 0.21 \text{ mA}$$

A small battery with the capacity of 1000 mAh has the possibility of power supply during:
 $1000 / 0.21 = 4761$ hours that is almost 200 days.


Even if we divide this result by 2 we still have 3 months of power supply.

CellCube integrates the **solar panel interface** (6-7V) for small panels with 100mv to 1W power output. These solar components provide much longer operation periods.

The CubeCell products support development with the Arduino framework. Sketches are uploaded via the serial port. The development boards have a USB port with USB-to-serial so sketches can be easily uploaded via USB. Uploading sketches to the sensor capsules requires a special adapter from Heltec.

In June 2020 supported for **PlatformIO** (from **Ukraine**) was added. Heltec uses a custom CubeCell bootloader for ASR650x. Serial number and a license that enables Arduino support are stored in flash memory.

1.2.1 Software installation with the use Arduino board manager



Open Arduino IDE, and click

File→Preferences→Settings

https://github.com/HelTecAutomation/CubeCell-Arduino/releases/download/V1.3.0/package_CubeCell_index.json

1.2.2 Software installation via Git

- **For Windows:**
<https://github.com/HelTecAutomation/ASR650x-Arduino/blob/master/InstallGuide/windows.md>
- **For MacOS:**
<https://github.com/HelTecAutomation/ASR650x-Arduino/blob/master/InstallGuide/mac.md>
- **For Linux (Ubuntu/Debian):** https://github.com/HelTecAutomation/ASR650x-Arduino/blob/master/InstallGuide/debian_ubuntu.md

Principal link for software resources:

<https://github.com/leroyle/ASR650x-Arduino>

1.3 Power consumption test with Power Profiler Kit II

The **Power Profiler Kit II** (PPK2) from **NORDIC** Semiconductor is a **standalone unit**, which can measure and optionally supply currents all the way from sub-uA and as high as 1A on all SmartComputerLab IoT DevKit boards from Smartcomputerlab.

The PPK2 is powered via a standard 5V USB cable, which can supply up to 500mA of current. In order to supply up to 1A of current, two USB cables are required.



Fig 1.4 Power Profiler Kit II

An **ampere meter only mode (to be used in our case)**, as well as a **source mode** (shown as AMP and source measure unit (SMU) respectively on the PCB) are supported. For the ampere meter mode, an external

power supply must source VCC levels between 0.8 and 5V to the device under test (DUT). For the source mode, the PPK2 supplies VCC levels between 0.8 and 5V and the on-board regulator supplies up to 1A of current to external applications.

It is possible to measure **low sleep** currents, the **higher active currents**, as well as **short current peaks**.

The PPK2 has an advanced analog measurement unit with a high dynamic measurement range. This allows accurate power consumption measurements for the entire range typically seen in low-power embedded applications, all the way from single μA s to 1A. The resolution varies between 100nA and 1mA depending on the measurement range and is high enough to detect small spikes as often seen in low power optimized systems.

The PPK2 can also use digital inputs as a low-end logic analyzer, enabling code-synchronized measurements. This can be achieved by connecting the digital inputs to an I/O pin on the external device under test (DUT). In order to use this functionality, the DUT must be powered by a VCC voltage of 1.6-5.5V. The digital input can then show what code is executed in the DUT at different points in time.

10x quicker sampling (i.e. 100ksps) compared with the long term window of the previous generation enables maximum continuous resolution at all times. This enables a user to gather average acquisition data and zoom in for high resolution data using the same window.



Fig 1.5 Screenshot of the **Power Profiler app** running in **ampere meter mode**.

The following figure shows the connection of the **IoT DevKit** with CubeCell board to the **Power Profiler Kit II** (PPK2). Note the use of JMP connector to provide **Vin/Vout** signal to the PPK2.




Fig 1.6 CubeCell IoT DevKit (on battery) connected to **PPK2** (jumper connector) and **solar panel**.

1.3.1 PPK2 software installation

The PPK2 software may be installed on Windows, MacOS and Linux OS. In our case we are using Ubuntu 20.04 LTS.

In order to install PPK2 software on Ubuntu we need 3 software components.

1. The **application** in executable mode:

```
nrfconnect-3.11.0-x86_64.appimage
```

To be found in:

<https://www.nordicsemi.com/Products/Development-tools/nRF-Connect-for-desktop/Download#infotabs>

2. The **Jlink** driver to be installed by:

```
sudo dpkg -i Jlink_Linux_V758b_x86_64.deb
```

To be found in: (look for **V758b** version):

<https://www.segger.com/downloads/jlink/>

3. The **nrf-udev** package to be installed by:

```
sudo dpkg -i nrf-udev_1.0.1-all.deb
```

To be found in:

<https://github.com/NordicSemiconductor/nrf-udev/releases>

1.4 Testing CubeCell HTCC-AB01 main board

In this initial labs we are going to test a number of essential **features** of the **HTCC-AB01** board. The knowledge of these features is necessary for the understanding of the following laboratories.

1.4.1 Board chip identifier

```
#include "Arduino.h"
void setup() {
  Serial.begin(9600);
  delay(100);
}

void loop() {
  Serial.println("in the loop");
  uint64_t chipID=getID();delay(100);
  Serial.println();Serial.println();
  Serial.printf("ChipID:%04X%08X\r\n", (uint32_t) (chipID>>32), (uint32_t) chipID);
  delay(1000);
}
```

1.4.2 Integrated RGB LED

The following is the basic code to test the integrated **RGB LED**.

```
#include "CubeCell_NeoPixel.h"
CubeCell_NeoPixel pixels(1, RGB, NEO_GRB + NEO_KHZ800);

void setup() {
  pinMode(Vext,OUTPUT);
  digitalWrite(Vext,LOW); //SET POWER
  pixels.begin(); // INITIALIZE NeoPixel strip object (REQUIRED)
  pixels.clear(); // Set all pixel colors to 'off'
}
uint8_t i=0;

void loop() {
  pixels.setPixelColor(0, pixels.Color(i, 0, 0));
  pixels.show(); // Send the updated pixel colors to the hardware.
  delay(200); // Pause before next pass through loop
  pixels.setPixelColor(0, pixels.Color(0, i, 0));
  pixels.show(); // Send the updated pixel colors to the hardware.
  delay(200); // Pause before next pass through loop
  pixels.setPixelColor(0, pixels.Color(0, 0, i));
  pixels.show(); // Send the updated pixel colors to the hardware.
  delay(200); // Pause before next pass through loop
  i+=10;
}
```

1.4.3 Integrated USR button

The following is the basic code to test the integrated **USR button**.

```
#include "Arduino.h"
#define USR GPIO7 // user button
uint32_t cnt = 0;

void cntIncrease()
{
  cnt++;
  Serial.println(cnt);
}

void setup() {
  Serial.begin(9600);
  PINMODE_INPUT_PULLUP(USR);
  attachInterrupt(USR, cntIncrease, FALLING);
}

void loop() {}
```


1.4.4 User flash memory

The chip has 1K user flash:

- the size of user flash row is 256;
- user flash row 0-2 can be edited;
- user flash row 3 is reserved, must not be edited;

```
#include "Arduino.h"
#define ROW 0
#define ROW_OFFSET 100

//CY_FLASH_SIZEOF_ROW is 256 , CY_SFLASH_USERBASE is 0xffff400
#define addr CY_SFLASH_USERBASE+CY_FLASH_SIZEOF_ROW*ROW + ROW_OFFSET

uint8_t data1[512];
uint8_t data2[512];
uint8_t data3;
uint8_t data4;
void setup() {
    Serial.begin(9600);delay(100);
    Serial.println();Serial.println();
    for(int i=0;i<512;i++){
        data1[i]=(uint8_t)i;
    }
    //write data1 to flash at addr
    FLASH_update(addr,data1,sizeof(data1));
    //read flash at addr to data2
    FLASH_read_at(addr,data2,sizeof(data2));
    uint16_t error=0;
    for(int i=0;i<512;i++){
        if(data1[i]!=data2[i])
        {
            Serial.printf("error:data1[%d] %d , data2[%d] %d \r\n",i,data1[i],i,data2[i]);
            error++;
        }
    }
    Serial.printf("error:%d\r\n",error);
    //read a byte at addr to data4
    FLASH_read_at(addr,&data4,1);
    Serial.printf("data4:%d\r\n",data4);
    data3=100;
    //write a byte at addr
    FLASH_update(addr,&data3,1);
    //read a byte at addr to data4
    FLASH_read_at(addr,&data4,1);
    Serial.printf("data4:%d\r\n",data4);
}

void loop() {
    // put your main code here, to run repeatedly:

}
```

The terminal printout:

```
error:0
data4:0
data4:100
```

1.4.5 Sleep modes: timer and user interruption

1.4.5.1 Timer mode

```
#include "Arduino.h"
#include "LoRaWan_APP.h"
#define timetillsleep 5000
#define timetillwakeUp 15000
static TimerEvent_t sleep;
static TimerEvent_t wakeUp;
uint8_t lowpower=1;

void onSleep()
{
  Serial.printf("Going into lowpower mode, %d ms later wake up.\r\n",timetillwakeUp);
  lowpower=1;
  //timetillwakeUp ms later wake up;
  TimerSetValue( &wakeUp, timetillwakeUp );
  TimerStart( &wakeUp );
}

void onWakeUp()
{
  Serial.printf("Woke up, %d ms later into lowpower mode.\r\n",timetillsleep);
  lowpower=0;
  //timetillsleep ms later into lowpower mode;
  TimerSetValue( &sleep, timetillsleep );
  TimerStart( &sleep );
}

void setup(){
  Serial.begin(9600);
  Radio.Sleep(); // LoRa modem sleep mode
  TimerInit( &sleep, onSleep );
  TimerInit( &wakeUp, onWakeUp );
  onSleep();
}

int i=0;

void loop() {
  if(lowpower){
    lowPowerHandler();
  }
  Serial.print('.');i++;
  if(i==40) { i=0; Serial.println();}
  delay(400);
}
```




Fig 1.7 PPK2 display for deep-sleep mode: max 13.04 mA (high-power), min 12.06 µA (low-power)

1.4.5.2 User interrupt mode

```
#include "Arduino.h"
#include "LoRa_APP.h"
#define INT_GPIO USER_KEY
#define timetillsleep 5000
static TimerEvent_t sleep;
uint8_t lowpower=1;

void onSleep()
{
  Serial.printf("Going into lowpower mode. Press user key to wake up\r\n");
  delay(5);
  lowpower=1;
}

void onWakeUp()
{
  delay(10);
  if(digitalRead(INT_GPIO) == 0)
  {
    Serial.printf("Woke up by GPIO, %d ms later into lowpower mode.\r\n",timetillsleep);
    lowpower=0;
    //timetillsleep ms later into lowpower mode;
    TimerSetValue( &sleep, timetillsleep );
    TimerStart( &sleep );
  }
}

void setup() {
  Serial.begin(9600);
  pinMode(INT_GPIO, INPUT);
  attachInterrupt (INT_GPIO,onWakeUp,FALLING);
  TimerInit( &sleep, onSleep );
  Serial.printf("Going into lowpower mode. Press user key to wake up\r\n");
  delay(5);
}

int i=0;

void loop() {
  if(lowpower){
    lowPowerHandler();
  }
  Serial.print('.');i++;
  if(i==40) { i=0; Serial.println();}
  delay(400);
}
```




Fig 1.8 PPK2 display for deep-sleep mode with user (USER button) interruption

1.4.6 System time

System time is base on internal timer counter that operates at milliseconds level (seconds/subseconds).

```
typedef struct TimerSysTime_s
{
    uint32_t Seconds;
    int16_t SubSeconds;
}
TimerSysTime_t;
```

The timer is operating also during the `deep_sleep` mode.

1.4.6.1 System time in active - high power mode

```
TimerSysTime_t sysTimeCurrent;
void setup()
{
    Serial.begin(9600);
    sysTimeCurrent = TimerGetSysTime( );
    Serial.printf("sys time:%u.%d\r\n", (unsigned int)sysTimeCurrent.Seconds,
sysTimeCurrent.SubSeconds);
    TimerSysTime_t newSysTime ;
    newSysTime.Seconds = 1000;
    newSysTime.SubSeconds = 50;
    TimerSetSysTime( newSysTime );
    sysTimeCurrent = TimerGetSysTime( );
    Serial.printf("sys time:%u.%d\r\n", (unsigned int)sysTimeCurrent.Seconds,
sysTimeCurrent.SubSeconds);
}

void loop() {
    delay(1000);
    sysTimeCurrent = TimerGetSysTime( );
    Serial.printf("sys time:%u.%d\r\n", (unsigned int)sysTimeCurrent.Seconds,
sysTimeCurrent.SubSeconds);
}
```

1.4.6.2 System time in deep_sleep - low power mode

```
#include "Arduino.h"
#include "LoRa_APP.h"
#include <Wire.h>
#include <SHT21.h>
#define timetillsleep 5000
#define timetillwakeUp 10000
static TimerEvent_t sleep;
static TimerEvent_t wakeUp;
uint8_t lowpower=1, highpower=0;
SHT21 sht;

float temp; // variable to store temperature
float humi; // variable to store hemidity

void getSHT21()
{
    pinMode(Vext, OUTPUT); digitalWrite(Vext, LOW);delay(50);
    Wire.begin();
    temp = sht.getTemperature(); // get temp from SHT
    humi = sht.getHumidity(); // get temp from SHT
    Serial.print("Humidity(%RH): ");Serial.print(humi);
    Serial.print("      Temperature(C): ");Serial.println(temp);
    Wire.end();
    digitalWrite(Vext, HIGH);
}

void onSleep()
{
    Serial.printf("Going into lowpower mode, %d ms later wake up.\r\n",timetillwakeUp);
    lowpower=1;highpower=0;
    TimerSetValue( &wakeUp, timetillwakeUp ); //timetillwakeUp ms later wake up;
    TimerStart( &wakeUp );
}
```

```

void onWakeUp()
{
    Serial.printf("Woke up, %d ms later into lowpower mode.\r\n",timetillsleep);
    lowpower=0;highpower=1;
    //timetillsleep ms later into lowpower mode;
    TimerSetValue( &sleep, timetillsleep );
    TimerStart( &sleep );
}

void setup() {
    Serial.begin(9600);
    Radio.Sleep( );
    TimerInit( &sleep, onSleep ); TimerInit( &wakeUp, onWakeUp );
    onSleep();
}

void loop() {
    if(lowpower){
        lowPowerHandler();
    }
    if(highpower)
        { getSHT21();highpower=0; }
}

```

Execution example:

```

Woke up, 5000 ms later into lowpower mode.
Humidity(%RH): 38.04      Temperature(C): 22.76
Going into lowpower mode, 10000 ms later w?Woke up, 5000 ms later into lowpower mode.
Humidity(%RH): 37.50      Temperature(C): 22.63
Going into lowpower mode, 10000 ms later w?Woke up, 5000 ms later into lowpower mode.
Humidity(%RH): 37.88      Temperature(C): 22.57
Going into lowpower mode, 10000 ms later w?Woke up, 5000 ms later into lowpower mode.
Humidity(%RH): 37.74      Temperature(C): 22.55

```

1.4.7 WatchDog timer

Watchdog timers are commonly found in [embedded systems](#) and other computer-controlled equipment where humans cannot easily access the equipment or would be unable to react to faults in a timely manner. In such systems, the computer cannot depend on a human to invoke a reboot if it [hangs](#); it must be self-reliant. For example, remote embedded systems such as [space probes](#) are not physically accessible to human operators; these could become permanently disabled if they were unable to autonomously recover from faults. In [robots](#) and other automated machines, a fault in the control computer could cause equipment damage or injuries before a human could react, even if the computer is easily accessed. A watchdog timer is usually employed in cases like these.

Watchdog timers are also used to monitor and limit software execution time on a normally functioning computer.

For example, a watchdog timer may be used when running untrusted code in a [sandbox](#), to limit the CPU time available to the code and thus prevent some types of [denial-of-service attacks](#).^[1] In [real-time operating systems](#), a watchdog timer may be used to monitor a time-critical task to ensure it completes within its maximum allotted time and, if it fails to do so, to terminate the task and report the failure.

```

#include "Arduino.h"
#include "innerWdt.h"

// For asr650x, the max feed time is 2.8 seconds.

#define MAX_FEEDTIME 2000 // default is 2800 ms

bool autoFeed = false;

void setup() {
    // put your setup code here, to run once:
    Serial.begin(9600);
    Serial.println();
    Serial.println("Start");

    /* Enable the WDT.
    * autoFeed = false: do not auto feed wdt.
    * autoFeed = true : it auto feed the wdt in every watchdog interrupt.
    */
}

```

```

    innerWdtEnable(autoFeed);
}

int feedCnt = 0;

void loop() {
    // put your main code here, to run repeatedly:
    Serial.println("running");
    delay(MAX_FEEDTIME - 100);

    if(autoFeed == false)
    {
        //feed the wdt
        if(feedCnt < 3)
        {
            Serial.println("feed wdt");
            feedInnerWdt();
            feedCnt++;
        }
        else
        {
            Serial.println("stop feed wdt");
        }
    }
}

```

To do

- Install the required software
- Test the above examples

1.5 External components and low power operation

In this lab we are building complete nodes using external components such as sensors, GPS modems, displays and relays. All these components are connected to the main board via **I2C**, **UART** bus or via simple logic lines. We start SSD1306 OLED display connected via I2C bus.




Fig 1.9 Low Power IoT DevKit from SmartComputerLab with CubeCell board

1.5.1 SSD1306 OLED



Fig 1.10 IoT DevKit with **OLED** screen connected via **I2C** bus

Note that the 3V3 pin is programmed via:

```
pinMode(Vext, OUTPUT);  
digitalWrite(Vext, LOW)
```

LOW state means that the output (3.3V) is active.

```
#include "LoRaWan_APP.h"  
#include "Arduino.h"  
#include "HT_SSD1306Wire.h"  
#include "Wire.h"  
SSD1306Wire display(0x3c, 500000, SDA, SCL, GEOMETRY_128_64, -1);  
  
void displayOLED(char *line1, char *line2, char *line3)  
{  
    display.init();  
    display.flipScreenVertically();display.clear();  
    display.drawString(20, 50, "SmartComputerLab");  
    display.drawString(0, 0, line1);  
    display.drawString(0, 15, line2);  
    display.drawString(0, 30, line3);  
    display.display();  
}
```



```

void setup() {
    Serial.begin(9600);
    pinMode(Vext, OUTPUT);
    digitalWrite(Vext, LOW);
    delay(100);
    Wire.begin(29,28);
    display.init();
    display.flipScreenVertically();
}

int c1=0,c2=0,c3=0;

void loop()
{
    char l1[32],l2[32],l3[32];
    sprintf(l1,"Count1=%d",c1);sprintf(l2,"Count2=%d",c2);
    sprintf(l3,"Count3=%d",c3);
    c1+=1;c2+=2;c3+=3;
    displayOLED(l1,l2,l3);
    delay(2000);
}

```

1.5.2 I2C device scan

The following code may be used to scan the I2C bus to find the connected devices (sensors,displays,...).

```

#include "Arduino.h"
#include "Wire.h"

void setup()
{
    Serial.begin(9600);
    pinMode(Vext,OUTPUT);
    digitalWrite(Vext,LOW);//set vext to high
    Wire.begin(); //(29,28);
}

void loop()
{
    byte error, address;
    int nDevices;
    Serial.println("Scanning...");
    nDevices = 0;
    for(address = 1; address < 127; address++)
    {
        Wire.beginTransmission(address);
        error = Wire.endTransmission();
        if (error == 0)
        {
            Serial.print("I2C device found at address 0x");
            if (address<16)
                Serial.print("0");
            Serial.print(address,HEX);
            Serial.println(" !");
            nDevices++;
        }
        else if (error==4)
        {
            Serial.print("Unknown error at address 0x");
            if (address<16)
                Serial.print("0");
            Serial.println(address,HEX);
        }
    }
    if (nDevices == 0)
        Serial.println("No I2C devices found\n");
    else
        Serial.println("done\n");
    delay(5000);
}

```

The terminal display for SSD1306 OLED on I2C bus (address - 0x3C)

..

```
Scanning...
I2C device found at address 0x3C !
done
```

```
Scanning...
I2C device found at address 0x3C !
done
```

```
Scanning...
I2C device found at address 0x3C !
done
```

1.5.3 SHT21 sensor module (I2C)

Driver github:

<https://github.com/e-radionica.com/SHT21-Arduino-Library>

An example sketch that reads the sensor and prints the relative humidity to the serial port


```
#include <SHT21.h> // include SHT21 library

SHT21 sht;

float temp; // variable to store temperature
float humidity; // variable to store humidity

void setup() {
  pinMode(Vext, OUTPUT);
  digitalWrite(Vext, LOW); delay(100);
  Wire.begin(); // begin Wire(I2C)
  Serial.begin(9600); // begin Serial
}

void loop() {
  temp = sht.getTemperature(); // get temp from SHT
  humidity = sht.getHumidity(); // get temp from SHT
  Serial.print("Temp: "); // print readings
  Serial.print(temp);
  Serial.print("\t Humidity: ");
  Serial.println(humidity);
  delay(85); // min delay
}
```



1.5.3.1 Reading SHT21 sensor and display on OLED

```
#include <SHT21.h> // include SHT21 library
#include "HT_SSD1306Wire.h"
SHT21 sht;
float temp; // variable to store temperature
float humi; // variable to store humidity

SSD1306Wire display(0x3c, 100000, SDA, SCL, GEOMETRY_128_64, -1);

void displayOLED(char *line1, char *line2, char *line3)
{
  Serial.println("in oled");
  display.init();
  display.flipScreenVertically(); display.clear();
  display.drawString(20, 50, "SmartComputerLab");
  display.drawString(0, 0, line1);
  display.drawString(0, 15, line2);
  display.drawString(0, 30, line3);
  display.display();
  delay(1000);
}

void setup() {
  pinMode(Vext, OUTPUT);
  digitalWrite(Vext, LOW); delay(100);
}
```

```

Wire.begin(); // begin Wire(I2C)
Serial.begin(9600); // begin Serial
}

void loop() {
char buff[32],bufft[32],buffh[32];
temp = sht.getTemperature(); // get temp from SHT
humid = sht.getHumidity(); // get temp from SHT

Serial.print("Temp: "); // print readings
Serial.print(temp);
Serial.print("\t Humidity: ");
Serial.println(humid);
sprintf(bufft, "T:%d", (int)temp); sprintf(buffh, "H:%d", (int)humid);
displayOLED(bufft,buffh, " ");
delay(85); // min delay for 14bit temp reading is 85ms
}

```

To do:

Transform the float into two integers for integer and fraction values.

```

dt=(int)((t-(int)t)*100.0); dh=(int)((h-(int)h)*100.0);
sprintf(buff, "T:%d.%d, H:%d.%d\n", (int)t, dt, (int)h, dh);
sprintf(bufft, "T:%d.%d", (int)t, dt);
sprintf(buffh, "H:%d.%d", (int)h, dh);

```

1.5.3.2 Raw SHT21 driver

The following code shows how operates the I2C bus in order to activate and to communicate with an SHT21 sensor. Note the control codes and the **Wire** (**Wire1**, **Wire2**) operations.

```

#define eSHT2xAddress 0x40
#define eTempHoldCmd 0xE3
#define eRHumidityHoldCmd 0xE5
#define eTempNoHoldCmd 0xF3
#define eRHumidityNoHoldCmd 0xF5

Wire.beginTransmission(eSHT2xAddress);
Wire.write(command);
Wire.endTransmission();
Wire.requestFrom(eSHT2xAddress, 3);
Wire.available()
Wire.read();

```

The complete code:

```

#define eSHT2xAddress 0x40
#define eTempHoldCmd 0xE3
#define eRHumidityHoldCmd 0xE5
#define eTempNoHoldCmd 0xF3
#define eRHumidityNoHoldCmd 0xF5

uint16_t readSensor(uint8_t command)
{
uint16_t result;
Wire.beginTransmission(eSHT2xAddress);
Wire.write(command);
Wire.endTransmission();
delay(100);
Wire.requestFrom(eSHT2xAddress, 3);
uint32_t timeout = millis() + 300; // Don't hang here for more than 300ms
while (Wire.available() < 3) {
if ((millis() - timeout) > 0) { return 0; }
}
//Store the result
result = Wire.read() << 8;
result += Wire.read();
result &= ~0x0003; // clear two low bits (status bits)
//Clear the final byte from the buffer
Wire.read();
return result;
}

```

```

float sht21_temperature(void)
{
    float value = readSensor(eTempHoldCmd);
    if (value == 0) {
        return -273; // Roughly Zero Kelvin indicates an error
    }
    return -46.85 + 175.72 / 65536.0 * value;
}

float sht21_humidity(void)
{
    float value = readSensor(eRHumidityHoldCmd);
    if (value == 0) {
        return 0; // Some unrealistic value
    }
    return -6.0 + 125.0 / 65536.0 * value;
}

```

To do:

1. Analyze and test the above raw driver
2. Replace the SHT21.h library with raw driver and test the new program code.
3. Use **PPK2** to analyze the power consumption , experiment with:


```
pinMode(Vext, OUTPUT);
digitalWrite(Vext, LOW); delay(100);
```

and

```
digitalWrite(Vext, HIGH); delay(100);
```

1.5.4 BH1750 – Ambient Light sensor

The BH1750 provides 16-bit light measurements in **lux**, the **SI** unit for measuring light making it easy to compare against other values like references and measurements from other sensors. The BH1750 is able to measure from 0 to **65K+** lux.


```

#include <Wire.h>
#include <Wire.h>
#include <BH1750.h>
BH1750 lightMeter;

void setup()
{
    Serial.begin(9600);
    pinMode(Vext, OUTPUT); delay(100);
    Wire.begin();
}

float lux;
void loop()
{
    digitalWrite(Vext, LOW); // start power before activating Wire
    delay(200);
    lightMeter.begin();
    delay(200);
    lux = lightMeter.readLightLevel();
    Serial.print("Light: ");
    Serial.print(lux);
    Serial.println(" lux");
    Wire.end(); // end Wire before disconnecting power
    digitalWrite(Vext, HIGH);
    delay(3000);
}

```



1.5.4.1 Raw BH1750 driver

The following is the BH1759 driver code allowing for direct integration on I2C bus with **Wire**, **Wire1**, **Wire2**, .

- `Wire.beginTransaction(0x23)` - address: start of transaction
- `Wire.write(commande);` - command on one byte
- `Wire.endTransmission();` - end of transaction
- `Wire.requestFrom(0x23, 2);` - number of requested bytes (data)
- `buff = Wire1.read();` - reading one byte

```
#include <Wire.h>

uint16_t readLightLevel(void)
{
    uint16_t level;
    Wire1.beginTransaction(0x23);
    Wire1.write(0x01); // power on
    Wire1.endTransmission();delay(50);
    Wire1.beginTransaction(0x23);
    Wire1.write(0x13); // high resolution - 120 ms, 0x13 - low resolution - 20 ms
    Wire1.endTransmission();
    delay(50); // high resolution - delay(200);
    Wire1.beginTransaction(0x23);
    Wire1.requestFrom( 0x23, 2);
    level = Wire1.read();
    level <= 8;
    level |= Wire1.read();
    Wire1.endTransmission();delay(40);
    level = level/1.2; // convert to lux
    return level;
}

void setup()
{
    Serial.begin(9600);
    pinMode(Vext, OUTPUT);
    digitalWrite(Vext, LOW);delay(50);
}


void loop()
{
    pinMode(Vext, OUTPUT);
    digitalWrite(Vext, LOW); delay(50);
    Wire1.begin();
    Serial.print("Luminosity(lux): ");
    Serial.println(readLightLevel());
    Wire1.end();
    digitalWrite(Vext, HIGH);
    delay(1000);
}
```

To do

1. Analyze and test the above example with **Wire** (not **Wire1** !)
2. Add the display on the OLED screen to show the luminosity value
3. Use **PPK2** to analyze the power consumption.

1.5.5 Air (CO2/TVOC) sensor CCS811 (SGP30)

CCS811 based sensor(I2C) VOC/eCO2 is an air quality monitoring sensor. This sensor is a gas sensor that can detect a wide range of Volatile Organic Compounds (VOCs) and is intended for indoor air quality monitoring. When connected to your micro-controller such as ARS6501 and running corresponding library code it will return a Total Volatile Organic Compound (TVOC) reading and an equivalent carbon dioxide reading (eCO2) over I2C.



The CCS811 has a 'standard' hot-plate MOX sensor, as well as a small micro-controller that controls power to the plate, reads the analog voltage, and provides an I2C interface to read from.

This part will measure eCO2 (equivalent calculated carbon-dioxide) concentration within a range of 400 to 8192 parts per million (ppm), and TVOC (Total Volatile Organic Compound) concentration within a range of 0 to 1187 parts per billion (ppb).

According to the fact sheet it can detect Alcohols, Aldehydes, Ketones, Organic Acids, Amines, Aliphatic and Aromatic Hydrocarbons.

Please note, this sensor, like all VOC/gas sensors, has variability and to get precise measurements you will want to calibrate it against known sources! That said, for general environmental sensors, it will give you a good idea of trends and comparisons.

Also, it is recommended that you run this sensor for 48 hours when you first receive it to "burn it in", and then 20 minutes in the desired mode every time the sensor is in use.

This is because the sensitivity levels of the sensor will change during early use.

The **CCS811** has a configurable interrupt pin that can fire when a conversion is ready and/or when a reading crosses a user-settable threshold.

The CCS811 supports multiple drive modes to take a measurement every 1 second, every 10 seconds, every 60 seconds, or every 250 milliseconds.

The device's I2C address is **0x5A**.

```
#include <Wire.h>
#include "Adafruit_SGP30.h"
Adafruit_SGP30 sgp;

uint32_t getAbsoluteHumidity(float temperature, float humidity) {
    // approximation formula from Sensirion SGP30 Driver Integration chapter 3.15
    const float absoluteHumidity = 216.7f * ((humidity / 100.0f) * 6.112f * exp((17.62f *
temperature) / (243.12f + temperature)) / (273.15f + temperature)); // [g/m^3]
    const uint32_t absoluteHumidityScaled = static_cast<uint32_t>(1000.0f * absoluteHumidity);
    return absoluteHumidityScaled;
}

void setup()
{
    Serial.begin(9600);
    pinMode(Vext, OUTPUT);
    digitalWrite(Vext, LOW); //set vext to high
    delay(100);
    Wire.begin(29, 28);
    Serial.println("SGP30 test");
    while (!sgp.begin()){
        Serial.print(".");delay(500);
    }
    Serial.print("Found SGP30 serial #");
    Serial.print(sgp.serialnumber[0], HEX);
    Serial.print(sgp.serialnumber[1], HEX);
    Serial.println(sgp.serialnumber[2], HEX);
    // If you have a baseline measurement from before you can assign it to start, to 'self-calibrate'
    //sgp.setIAQBaseline(0x8E68, 0x8F41); // Will vary for each sensor!
}
```

```

int counter = 0;
void loop() {
  if (!sgp.IAQmeasure()) {
    Serial.println("Measurement failed");
    return;
  }
  Serial.print("TVOC "); Serial.print(sgp.TVOC); Serial.print(" ppb\t");
  Serial.print("eCO2 "); Serial.print(sgp.eCO2); Serial.println(" ppm");

  if (!sgp.IAQmeasureRaw()) {
    Serial.println("Raw Measurement failed");
    return;
  }
}
}

```

To do

1. Test the above example
2. Add the display on the OLED screen to show the **TVOC** and **CO2** values
3. use `digitalWrite(Vext, LOW);` and `digitalWrite(Vext, HIGH);` to set on and off the power line `Vext` in the `loop()` function (**notice the problem !**)

1.5.6 “Time of Flight” distance sensors – VL53L0X and VL53L1X

1.5.6.1 VL53L03

The **VL53L0X** contains a very tiny invisible laser source, and a matching sensor. The **VL53L0X** can detect the "time of flight", or how long the light has taken to bounce back to the sensor. Since it uses a very narrow light source, it is good for determining distance of only the surface directly in front of it. Unlike sonars that bounce ultrasonic waves, the 'cone' of sensing is very narrow. Unlike IR distance sensors that try to measure the amount of light bounced, the VL53L0x is much more precise and doesn't have linearity problems or 'double imaging' where you can't tell if an object is very far or very close.



```

#include "Adafruit_VL53L0X.h"

Adafruit_VL53L0X lox = Adafruit_VL53L0X();

void setup() {
  pinMode(Vext, OUTPUT);
  Serial.begin(9600);
  Serial.println("Adafruit VL53L0X test");
  Serial.println(F("VL53L0X API Simple Ranging example\n\n"));
}

void loop() {
  digitalWrite(Vext, LOW);
  delay(50);
  Wire.begin(29, 28);
  if (!lox.begin()) {
    Serial.println(F("Failed to boot VL53L0X"));
    digitalWrite(Vext, HIGH);
    delay(1000);
    return;
  }


  Serial.print("Reading a measurement... ");
  VL53L0X_RangingMeasurementData_t measure;
  lox.rangingTest(&measure, false); // pass in 'true' to get debug data printout!
  Wire.end();
  digitalWrite(Vext, HIGH);
  if (measure.RangeStatus != 4) { // phase failures have incorrect data
    Serial.print("Distance (mm): ");
    Serial.println(measure.RangeMilliMeter);
  } else {
    Serial.println(" out of range ");
  }
  delay(1000);
}

```


1.5.6.2 VL53L1X

This example demonstrates how to read and average distance, the measurement status, and the signal rate. Use long distance mode and allow up to 50000 us (50 ms) for a measurement.

You can change these settings to adjust the performance of the sensor, but the minimum timing budget is 20 ms for short distance mode and 33 ms for medium and long distance modes.



```
sensor.setDistanceMode(VL53L1X::Long);  
sensor.setMeasurementTimingBudget(50000);
```

Start continuous readings at a rate of one measurement every 50 ms (the inter-measurement period). This period should be at least as long as the timing budget.

```
sensor.startContinuous(50);
```

The complete code:

```
#include "Arduino.h"  
#include <Wire.h>  
#include "VL53L1X.h"  
  
VL53L1X sensor;  
  
void setup(void)  
{  
  pinMode(Vext, OUTPUT);  
  digitalWrite(Vext, LOW);  
  delay(500);  
  Serial.begin(9600);  
  Serial.println(); Serial.println(); delay(100);  
  Wire.begin(29,28);  
  Wire.setClock(400000); // use 400 kHz I2C  
  sensor.setTimeout(500);  
  Serial.println("before init");  
  if (!sensor.init())  
  {  
    Serial.println("Failed to detect and initialize sensor!");  
    while (1);  
  }  
  sensor.setDistanceMode(VL53L1X::Long);  
  sensor.setMeasurementTimingBudget(50000);  
  sensor.startContinuous(50);  
}  
  
void loop()  
{  
  Serial.print(sensor.read());  
  if (sensor.timeoutOccurred()) { Serial.print(" TIMEOUT"); }  
  Serial.println();  
}
```

To do

- Test the above examples of **VL53LXX** sensors
- Add the display on the OLED screen to show the distance value

1.5.7 SoftwareSerial – GPS - ATGM33

This sample code demonstrates the normal use of a **TinyGPS++ (TinyGPSPlus)** object. It requires the use of **SoftwareSerial**, and assumes that you have a 9600-baud serial GPS device hooked up on pins **GPIO3** - (rx) and **GPIO5** - (tx).



3.5.7.1 Simple – direct UART stream

```
#include <Arduino.h>
#include <softSerial.h>
softSerial softwareSerial(GPIO5 /*TX pin*/, GPIO3 /*RX pin*/);
// GPIO5 (33) , GPIO3 (8)

void setup()
{
  pinMode(Vext, OUTPUT);
  digitalWrite(Vext, LOW);
  delay(500);
  Serial.begin(9600);
  softwareSerial.begin(9600);
  delay(1000);
  Serial.println("Normal serial init");
}
char *ptr, gmt[12], clarg[12], clong[12];

void loop()
{
  if(softwareSerial.available())
  {
    char serialbuffer[256] = {0};
    int i = 0;
    while (softwareSerial.available() && i<256)
    {
      serialbuffer[i] = (char)softwareSerial.read();
      i++;
    }
    serialbuffer[i] = '\0';
    if(serialbuffer[0])
    {
      Serial.println(serialbuffer);
      ptr=strstr(serialbuffer,"RMC,");
      strncpy(gmt,ptr+4,6); Serial.print("GMT:");Serial.println(gmt);
      ptr=strstr(serialbuffer,"A,");
      strncpy(clarg,ptr+3,10); Serial.print("Larg:");Serial.println(clarg);
      ptr=strstr(serialbuffer,"N,");
      strncpy(clong,ptr+3,10); Serial.print("Long:");Serial.println(clong);
      Serial.println();
    }
  }
  delay(200);
}
```

1.5.7.2 Simple – direct UART stream and OLED display

```
#include "LoRaWan_APP.h"
#include "Arduino.h"
#include "HT_SSD1306Wire.h"
#include "Wire.h"
#include <softSerial.h>
softSerial softwareSerial(GPIO5 /*TX pin*/, GPIO3 /*RX pin*/); // GPIO5 (33) , GPIO3 (8)
SSD1306Wire display(0x3c, 500000, SDA, SCL, GEOMETRY_128_64, -1);
// addr , freq , i2c group , resolution , rst

void displayOLED(char *line1, char *line2, char *line3)
{
  display.init();
  display.flipScreenVertically();display.clear();
  display.drawString(20, 50, "SmartComputerLab" );
  display.drawString(0, 0, line1 );
  display.drawString(0, 15, line2);
  display.drawString(0, 30, line3);
  display.display();
}
```

```

void setup()
{
  pinMode(Vext, OUTPUT);
  digitalWrite(Vext, LOW);
  delay(500);
  Serial.begin(9600);
  Wire.begin(29,28);
  softwareSerial.begin(9600);
  delay(1000);
  Serial.println("Normal serial init");
}
char *ptr, gmt[12], clarg[12], clong[12], dgmt[24], dclarg[24], dclong[24];
void loop()
{
  if(softwareSerial.available())
  {
    char serialbuffer[256] = {0};
    int i = 0;
    while (softwareSerial.available() && i<256)
    {
      serialbuffer[i] = (char)softwareSerial.read();i++;
    }
    serialbuffer[i] = '\0';
    if(serialbuffer[0])
    {
      //Serial.print("Received data from software Serial:");
      Serial.println(serialbuffer);
      ptr=strstr(serialbuffer,"RMC,");
      strncpy(gmt,ptr+4,6); Serial.print("GMT:");Serial.println(gmt);
      ptr=strstr(serialbuffer,"A,");
      strncpy(clarg,ptr+3,10); Serial.print("Larg:");Serial.println(clarg);
      ptr=strstr(serialbuffer,"N,");
      strncpy(clong,ptr+3,10); Serial.print("Long:");Serial.println(clong);
      Serial.println();
      sprintf(dgmt,"GMT:%s",gmt);sprintf(dclarg,"LAT:%s",clarg);sprintf(dclong,"LNG:%s (W)",clong);
      displayOLED(dgmt, dclarg, dclong);
    }
  }
  delay(600);
}

```

1.5.7.3 UART stream decoded by TinyGPS++ library

```

#include <TinyGPS++.h>
#include <softSerial.h>
static const uint32_t GPSPBaud = 9600;
TinyGPSPlus gps;
softSerial ss(GPIO5 /*TX pin*/, GPIO3 /*RX pin*/); // GPIO5 (33) , GPIO3 (8)

static void smartDelay(unsigned long ms)
{
  unsigned long start = millis();
  do
  {
    while (ss.available())
      gps.encode(ss.read());
  } while (millis() - start < ms);
}

static void printFloat(float val, bool valid, int len, int prec)
{
  if (!valid)
  {
    while (len-- > 1) Serial.print('*'); Serial.print(' ');
  }
  else
  {
    Serial.print(val, prec);
    int vi = abs((int)val);
    int flen = prec + (val < 0.0 ? 2 : 1); // . and -
    flen += vi >= 1000 ? 4 : vi >= 100 ? 3 : vi >= 10 ? 2 : 1;
    for (int i=flen; i<len; ++i) Serial.print(' ');
  }
  smartDelay(0);
}

```

```

static void printInt(unsigned long val, bool valid, int len)
{
    char sz[32] = "*****";
    if (valid) sprintf(sz, "%ld", val);
    sz[len] = 0;
    for (int i=strlen(sz); i<len; ++i) sz[i] = ' ';
    if (len > 0) sz[len-1] = ' ';
    Serial.print(sz);
    smartDelay(0);
}

static void printDateTime(TinyGPSTime &t, TinyGPSDate &d)
{
    if (!d.isValid()) { Serial.print(F("***** ")); }
    else
    {
        char sz[32];
        sprintf(sz, "%02d/%02d/%02d ", d.month(), d.day(), d.year());
        Serial.print(sz);
    }
    if (!t.isValid())
    {
        Serial.print(F("***** ")); }
    else
    {
        char sz[32];
        sprintf(sz, "%02d:%02d:%02d ", t.hour(), t.minute(), t.second());
        Serial.print(sz);
    }
    printInt(d.age(), d.isValid(), 5);
    smartDelay(0);
}

static void printStr(const char *str, int len)
{
    int slen = strlen(str);
    for (int i=0; i<len; ++i)
        Serial.print(i<slen ? str[i] : ' ');
    smartDelay(0);
}

void setup()
{
    pinMode(Vext, OUTPUT);
    digitalWrite(Vext, LOW);
    delay(500);

    Serial.begin(9600);
    ss.begin(GPSBaud);

    Serial.println(F("FullExample.ino"));
    Serial.println(F("An extensive example of many interesting TinyGPS++ features"));
    Serial.print(F("Testing TinyGPS++ library v. ")); Serial.println(TinyGPSPlus::libraryVersion());
    Serial.println(F("by Mikal Hart"));
    Serial.println();
    Serial.println(F("Sats HDOP Latitude Longitude Fix Date Time Date Alt Course"));
    Serial.println(F("Speed Card Distance Course Card Chars Sentences Checksum"));
    Serial.println(F(" (deg) (deg) Age Age (m) --- from"));
    Serial.println(F("GPS ---- ---- to London ---- RX RX Fail"));

    Serial.println(F("-----"));
    Serial.println(F("-----"));
}

void loop()
{
    static const double LONDON_LAT = 51.508131, LONDON_LON = -0.128002;

    printInt(gps.satellites.value(), gps.satellites.isValid(), 5);
    printFloat(gps.hdop.hdop(), gps.hdop.isValid(), 6, 1);
    printFloat(gps.location.lat(), gps.location.isValid(), 11, 6);
    printFloat(gps.location.lng(), gps.location.isValid(), 12, 6);
    printInt(gps.location.age(), gps.location.isValid(), 5);
    printDateTime(gps.date, gps.time);
    printFloat(gps.altitude.meters(), gps.altitude.isValid(), 7, 2);
    printFloat(gps.course.deg(), gps.course.isValid(), 7, 2);

```

```

printFloat(gps.speed.kmph(), gps.speed.isValid(), 6, 2);
printStr(gps.course.isValid() ? TinyGPSPlus::cardinal(gps.course.deg()) : "*** ", 6);

unsigned long distanceKmToLondon =
  (unsigned long)TinyGPSPlus::distanceBetween(
    gps.location.lat(),
    gps.location.lng(),
    LONDON_LAT,
    LONDON_LON) / 1000;
printInt(distanceKmToLondon, gps.location.isValid(), 9);

double courseToLondon =
  TinyGPSPlus::courseTo(
    gps.location.lat(),
    gps.location.lng(),
    LONDON_LAT,
    LONDON_LON);
printFloat(courseToLondon, gps.location.isValid(), 7, 2);

const char *cardinalToLondon = TinyGPSPlus::cardinal(courseToLondon);

printStr(gps.location.isValid() ? cardinalToLondon : "*** ", 6);

printInt(gps.charsProcessed(), true, 6);
printInt(gps.sentencesWithFix(), true, 10);
printInt(gps.failedChecksum(), true, 9);
Serial.println();
smartDelay(1000);
if (millis() > 5000 && gps.charsProcessed() < 10)
  Serial.println(F("No GPS data received: check wiring"));
}

```

Terminal output fragment:

```

..
d 1.0 47.216858 -1.693524 d 07/04/2021 16:49:28 d 34.90 0.50 0.39 N d 12.77 NNE d d d
d 1.0 47.216862 -1.693527 d 07/04/2021 16:49:29 d 34.40 0.50 0.57 N d 12.77 NNE d d d
d 1.0 47.216862 -1.693528 d 07/04/2021 16:49:30 d 34.40 0.50 0.30 N d 12.77 NNE d d d
d 1.0 47.216858 -1.693528 d 07/04/2021 16:49:31 d 34.00 0.50 0.33 N d 12.77 NNE d d d
d 1.0 47.216858 -1.693528 d 07/04/2021 16:49:32 d 33.90 0.50 0.41 N d 12.77 NNE d d d
d 1.0 47.216862 -1.693529 d 07/04/2021 16:49:34 d 33.90 0.38 0.24 N d 12.77 NNE d d d
d 1.0 47.216866 -1.693530 d 07/04/2021 16:49:35 d 32.90 0.38 0.74 N d 12.77 NNE d d d
d 1.0 47.216866 -1.693527 d 07/04/2021 16:49:36 d 32.50 0.38 0.72 N d 12.77 NNE d d d
d 1.0 47.216866 -1.693526 d 07/04/2021 16:49:37 d 31.80 0.38 0.61 N d 12.77 NNE d d d
d 1.0 47.216862 -1.693525 d 07/04/2021 16:49:38 d 32.30 0.38 0.59 N d 12.77 NNE d d d
d 1.0 47.216862 -1.693527 d 07/04/2021 16:49:39 d 32.00 0.38 0.13 N d 12.77 NNE d d d
d 1.0 47.216862 -1.693526 d 07/04/2021 16:49:41 d 32.40 0.38 0.61 N d 12.77 NNE d d d
d 1.0 47.216862 -1.693528 d 07/04/2021 16:49:42 d 32.20 0.38 0.07 N d 12.77 NNE d d d
d 1.0 47.216858 -1.693526 d 07/04/2021 16:49:43 d 31.80 0.38 0.69 N d 12.77 NNE d d d
d 1.0 47.216858 -1.693527 d 07/04/2021 16:49:44 d 32.40 0.38 0.22 N d 12.77 NNE d d d

```

To do

- Test the above examples of **GPS** receivers
- Add the display on the OLED screen to show the value of time, longitude, and latitude for the second example code

1.6 Low power operation period with `lowPowerHandler()`

In this lab we have already introduced low power technique to read and display the data on sensors and OLED screen with cutting off the power line during the idle period.

Now we add the **low_power** periods for the operation of the processor (MCU) itself.

To start let's study the following example.

1.6.1 Low power operation with SHT21 sensor

The first part of the code adds the necessary libraries and the declaration of the sensor (SHT21).

```
#include "Arduino.h"
#include <Wire.h>
#include "SHT21.h"
SHT21 sht;
```

The second fragment contains the definition and the declaration of the constants and variables used to configure the events required for low and high power operation (periods).

```
#define timetillsleep 5000
#define timetillwakeUp 10000
static TimerEvent_t sleep;
static TimerEvent_t wakeUp;
uint8_t lowpower=1, highpower=0;
```

`timetosleep` and `timetillwakeUp` define the periods of high and low power operation. In this case the board will be set for low power period of 10 secs and high power period of 5 secs.

The corresponding timer events are `sleep` for the high-to-low power period and `wakeUp` for the low-to-high power period.

`lowpower` and `highpower` variables keep the actual state of the operation.

The next section of the code defines the **ISR functions** activated by the time on `sleep` and on `wakeUp` events. They are:

```
void onSleep()
{
    Serial.printf("Go to lowpower mode,%d ms later wake up.\r\n",timetillwakeUp);
    lowpower=1;highpower=0; //
    //timetillwakeUp ms later wake up;
    TimerSetValue(&wakeUp,timetillwakeUp);
    TimerStart(&wakeUp);
}

void onWakeUp()
{
    Serial.printf("Woke up, %d ms later into lowpower mode.\r\n",timetillsleep);
    lowpower=0;highpower=1;
    //timetillsleep ms later into lowpower mode;
    TimerSetValue(&sleep,timetillsleep);
    TimerStart(&sleep);
}
```

The setup section of the code initializes the pointers to `onSleep` and on `WakeUp` functions. The radio modem (LoRa) is set into sleep mode by `Radio.Sleep()`.

The execution of the program starts by `onSleep()` function in order to set the application into initial low power mode.

```
void setup()
{
    Serial.begin(9600);
    Radio.Sleep();
    TimerInit(&sleep, onSleep);
    TimerInit(&wakeUp, onWakeUp);
    onSleep();
}
```

The main `loop()` of the program oscillates between the low-power and high-power modes.

During the high-power period the `getSHT21()` function is called – **only once**.

```
void loop() {
  if(lowpower){
    lowPowerHandler();
  }
  if(highpower)
    { getSHT21();highpower=0; }
```

The last part of the code contains the `getSHT21()` function.

The execution of this function starts by the initialization of the power (3V3) line – **Vext** and by the activation of the I2C bus by `Wire.begin(29, 28)` ; then we activate the sensor to read the temperature and humidity data.

After the data readings the I2C bus is deactivated – `Wire.end()` and the power line is cutoff - `digitalWrite(Vext, HIGH)`.

```
float temp; // variable to store temperature
float humi; // variable to store hemidity
int dt,dh;
char buff[32];

void getSHT21()
{
  pinMode(Vext, OUTPUT);
  digitalWrite(Vext, LOW);
  delay(50);
  Wire.begin();
  temp = sht.getTemperature(); // get temp from SHT
  humi = sht.getHumidity(); // get temp from S
  Serial.print("Humidity(%RH): ");
  Serial.print(humi);
  Serial.print("      Temperature(C): ");
  Serial.println(temp);
  Wire.end();
  digitalWrite(Vext, HIGH);
}
```

The complete code is given below:

```
#include "Arduino.h"
#include "LoRa_APP.h"
#include <Wire.h>
#include <SHT21.h>
#define timetillsleep 5000
#define timetillwakeup 10000
static TimerEvent_t sleep;
static TimerEvent_t wakeUp;
uint8_t lowpower=1, highpower=0;
SHT21 sht;

float temp; // variable to store temperature
float humi; // variable to store hemidity

void getSHT21()
{
  pinMode(Vext, OUTPUT); digitalWrite(Vext, LOW);delay(50);
  Wire.begin();
  temp = sht.getTemperature(); // get temp from SHT
  humi = sht.getHumidity(); // get temp from SHT
  Serial.print("Humidity(%RH): ");Serial.print(humi);
  Serial.print("      Temperature(C): ");Serial.println(temp);
  Wire.end();
  digitalWrite(Vext, HIGH);
}

void onSleep()
{
  Serial.printf("Going into lowpower mode, %d ms later wake up.\r\n",timetillwakeup);
  lowpower=1;highpower=0;
  TimerSetValue( &wakeUp, timetillwakeup ); //timetillwakeup ms later wake up;
  TimerStart( &wakeUp );
}
```



```

void onWakeUp()
{
    Serial.printf("Woke up, %d ms later into lowpower mode.\r\n",timetillsleep);
    lowpower=0;highpower=1;
    //timetillsleep ms later into lowpower mode;
    TimerSetValue( &sleep, timetillsleep );
    TimerStart( &sleep );
}

void setup() {
    Serial.begin(9600);
    Radio.Sleep( );
    TimerInit( &sleep, onSleep ); TimerInit( &wakeUp, onWakeUp );
    onSleep();
}

void loop() {
    if(lowpower){
        lowPowerHandler();
    }
    if(highpower)
        { getSHT21();highpower=0; }
}

```



Fig 1.12 The resulting current consumption values from the integrated LiPo battery are:

- **11 mA** for **high power** operations, except the sensor capture (**50 mA**)
- **79.7 µA** for **low power** state

To do

1. Apply the same low/high power operational mode for other sensors.
2. Use simplified SHT21 driver with low precision option (reading time : 10ms)

1.6.2 Direct GPS - UART stream and OLED display with low_power mode

```
#include "LoRaWan_APP.h"
#include "Arduino.h"
#include "HT_SSD1306Wire.h"
#include "Wire.h"
#include <softSerial.h>
softSerial softwareSerial(GPIO5 /*TX pin*/, GPIO3 /*RX pin*/); // GPIO5 (33) , GPIO3 (8)

SSD1306Wire display(0x3c, 500000, SDA, SCL, GEOMETRY_128_64, -1);
// addr , freq , i2c group , resolution , rst

#define timetillsleep 5000
#define timetillwakeup 10000
static TimerEvent_t sleep;
static TimerEvent_t wakeUp;
uint8_t lowpower=1, highpower=0;

void displayOLED(char *line1, char *line2, char *line3)
{
    display.init();
    display.flipScreenVertically();display.clear();
    display.drawString(20, 50, "SmartComputerLab" );
    display.drawString(0, 0, line1 );
    display.drawString(0, 15, line2);
    display.drawString(0, 30, line3);
    display.display();
}

void onSleep()
{
    Serial.printf("Going into lowpower mode, %d ms later wake up.\r\n",timetillwakeup);
    lowpower=1;highpower=0;
    //timetillwakeup ms later wake up;
    TimerSetValue( &wakeUp, timetillwakeup );
    TimerStart( &wakeUp );
}

void onWakeUp()
{
    Serial.printf("Woke up, %d ms later into lowpower mode.\r\n",timetillsleep);
    lowpower=0;highpower=1;
    pinMode(Vext, OUTPUT);
    digitalWrite(Vext, LOW);
    softwareSerial.begin(9600);
    //timetillsleep ms later into lowpower mode;
    TimerSetValue( &sleep, timetillsleep );
    TimerStart( &sleep );
}

void setup()
{
    pinMode(Vext, OUTPUT);
    digitalWrite(Vext, LOW);
    delay(500);
    Serial.begin(9600);
    Wire.begin(29,28);
    softwareSerial.begin(9600);
    delay(1000);
    Serial.println("Normal serial init");
    TimerInit( &sleep, onSleep );
    TimerInit( &wakeUp, onWakeUp );
    onSleep();
}

char *ptr, gmt[12],clarg[12], clong[12],dgmt[24],dclarg[24], dclong[24];

void loop()
{
    if(lowpower){
        lowPowerHandler();
    }
    if(highpower)
    {

        if(softwareSerial.available())
        {
            char serialbuffer[256] = {0};
```

```

int i = 0;
while (softwareSerial.available() && i<256)
{
    serialbuffer[i] = (char)softwareSerial.read();
    i++;
}
serialbuffer[i] = '\0';
if(serialbuffer[0])
{
    //Serial.print("Received data from software Serial:");
    Serial.println(serialbuffer);
    ptr=strstr(serialbuffer,"RMC,");
    strncpy(gmt,ptr+4,6); Serial.print("GMT:");Serial.println(gmt);
    ptr=strstr(serialbuffer,"A,");
    strncpy(clarg,ptr+3,10); Serial.print("Larg:");Serial.println(clarg);
    ptr=strstr(serialbuffer,"N,");
    strncpy(dlong,ptr+3,10); Serial.print("Long:");Serial.println(dlong);
    Serial.println();
    sprintf(dgmt,"GMT:%s",gmt);sprintf(dclarg,"LAT:%s",clarg);sprintf(dclong,"LNG:%s
(W)",dlong);
    displayOLED(dgmt, dclarg, dclong);
}
Wire.end();
digitalWrite(Vext, HIGH);
}
}
//digitalWrite(Vext, HIGH);
delay(4000);
}

```



Fig 1.13 GPS read and display with low and high power modes

Lab 2

LoRa modem - basics and low power operation

The following examples show how to use the integrated LoRa modem (**SX1262**) in basic mode (pure LoRa). These examples take into account low power requirements including implemented through MCU **low_power** state **radio sleep** and the **power cutoff** for the external components (sensors, modems, displays, ..)

2.1 LoRa send with battery state value

In this first example of LoRa based communication we **send simple LoRa packets** using the functions provided in **LoRaWan_APP.h** library. The packets contain the **value of the battery state** provided in **mV**. LoRa radio channel is configured with several parameters including:

```
#define RF_FREQUENCY                868000000 // Hz
#define TX_OUTPUT_POWER            14         // dBm - max 21 dBm
#define LORA_BANDWIDTH              0         // [0: 125 kHz,
                                              // 1: 250 kHz,
                                              // 2: 500 kHz,
                                              // 3: Reserved]
#define LORA_SPREADING_FACTOR      8         // [SF7..SF12]
#define LORA_CODINGRATE             4         // [1: 4/5,
                                              // 2: 4/6,
                                              // 3: 4/7,
                                              // 4: 4/8]
#define LORA_PREAMBLE_LENGTH       8         // Same for Tx and Rx
#define LORA_SYMBOL_TIMEOUT         0         // Symbols
#define LORA_FIX_LENGTH_PAYLOAD_ON false
#define LORA_IQ_INVERSION_ON        false
```

The **LoRaWan_APP.h** library includes :

```
Radio.Send( (uint8_t *)txPacket, strlen(txPacket) );
```

to send a packet; and

```
RadioEvents.RxDone = OnRxDone;
Radio.Init( &RadioEvents );
```

```
void OnRxDone(uint8_t *payload,uint16_t size,int16_t rssi,int8_t snr ) { }
```

to receive Lora packets.

```
Radio.Sleep();
```

Puts the LoRa modem into **deep sleep** state. The activated modem requires at least **10 mA** to stay awake.

Our first code example starts with the configuration of the radio parameters including: the frequency, the signal bandwidth, the spreading factor, the coding rate, etc.

At the receiving side we need to use the same parameters.

The integrated RGB LED needs about **5 mA** to be active , if you may deactivate the LED with:

```
turnOffRGB();
```

To activate the LED you may use:

```
turnOnRGB(COLOR_SEND,0); - red
turnOnRGB(COLOR_RECEIVED,0); - green
```

In order to keep the power consumption during the passive periods as low as possible you have to use these 3 functions:

- `lowPowerHandler();`
- `Radio.Sleep();`
- `turnOffRGB();`

2.1.1 Sending data packet (battery voltage)

This first example of LoRa sender implements only some low power features , the code just sends the value of the battery voltage.

```
#include "LoRaWan_APP.h"
#include "Arduino.h"
#ifdef LoraWan_RGB
#define LoraWan_RGB 0
#endif
#define RF_FREQUENCY 868500000 // Hz
#define TX_OUTPUT_POWER 14 // dBm
#define LORA_BANDWIDTH 0 // [0: 125 kHz,
                          // 1: 250 kHz,
                          // 2: 500 kHz,
                          // 3: Reserved]
#define LORA_SPREADING_FACTOR 9 // [SF7..SF12]
#define LORA_CODINGRATE 1 // [1: 4/5,
                          // 2: 4/6,
                          // 3: 4/7,
                          // 4: 4/8]
#define LORA_PREAMBLE_LENGTH 8 // Same for Tx and Rx
#define LORA_SYMBOL_TIMEOUT 0 // Symbols
#define LORA_FIX_LENGTH_PAYLOAD_ON false
#define LORA_IQ_INVERSION_ON false
#define RX_TIMEOUT_VALUE 1000
#define BUFFER_SIZE 128 // Define the payload size here

char txPacket[BUFFER_SIZE];
static RadioEvents_t RadioEvents;
void OnTxDone( void );
void OnTxTimeout( void );

typedef enum
{
    LOWPOWER, ReadVoltage, TX // 3 states (1,2,3)
} States_t;

States_t state;
bool sleepMode = false;
int16_t rssi, rxSize;
uint16_t voltage;

void setup()
{
    Serial.begin(9600);
    voltage = 0;
    rssi=0;
    RadioEvents.TxDone = OnTxDone;
    RadioEvents.TxTimeout = OnTxTimeout;
    Radio.Init( &RadioEvents );
    Radio.SetChannel( RF_FREQUENCY );
    Radio.SetTxConfig( MODEM_LORA, TX_OUTPUT_POWER, 0, LORA_BANDWIDTH,
                      LORA_SPREADING_FACTOR, LORA_CODINGRATE,
                      LORA_PREAMBLE_LENGTH, LORA_FIX_LENGTH_PAYLOAD_ON,
                      true, 0, 0, LORA_IQ_INVERSION_ON, 3000 );

    state=ReadVoltage;
}

void loop()
{
    switch(state)
    {
        case TX:
        {
            memset(txPacket, 0x00, BUFFER_SIZE);
            sprintf(txPacket, "%s", "ADC_battery (mV): ");
            int plen= strlen(txPacket);
            sprintf(txPacket+plen, "%d", voltage);
            if(voltage<(uint16_t)3680)turnOnRGB(COLOR_SEND,0);
            else turnOnRGB(COLOR_RECEIVED,200);
            Serial.printf("\r\nsending packet \"%s\"\r\n", txPacket);
            Radio.Send( (uint8_t *)txPacket, strlen(txPacket) );
            Serial.println(strlen(txPacket));delay(100);
            state=LOWPOWER;
        }
    }
}
```

```

        break;
    }
    case LOWPOWER:
    {
        lowPowerHandler();delay(100);
        turnOffRGB();
        delay(2000); //LowPower time
        state = ReadVoltage;
        break;
    }
    case ReadVoltage:
    {
        pinMode(VBAT_ADC_CTL,OUTPUT);
        digitalWrite(VBAT_ADC_CTL,LOW);
        voltage=analogRead(ADC)+550; /*2;
        pinMode(VBAT_ADC_CTL, INPUT);
        state = TX;
        break;
    }
    default:
        break;
}
Radio.IrqProcess();
}

void OnTxDone( void )
{
    Serial.print("TX done!");
    turnOnRGB(0,0);
}

void OnTxTimeout( void )
{
    Radio.Sleep( );
    Serial.print("TX Timeout.....");
    state=ReadVoltage;
    Serial.print(state);
}

```




Fig 2.1 Sending three LoRa packets

To do

1. Analyze and test the above example

2.2 Sending data packet with low_power management

In the following example we introduce the element of power management. During the **passive period** we downgrade the activity of the MCU with `lowPowerHandler()`, we suspend the activity of the LoRa modem with `Radio.Sleep()`, and we deactivate integrated LED with `turnOffRGB()`.

These preparations allow us to reduce the power consumption (current) to about **20 μ A** during the passive periods defined as - `timetillwakeUp 15000`.

During the active period defined by - `timetillsleep 1000`, the voltage capture and the LoRa packet transmission, power consumption goes up to **60 mA or more** depending on the transmission power `TX_OUTPUT_POWER` parameter.

```
#include "LoRaWan_APP.h"
#include "Arduino.h"
#define timetillsleep 1000
#define timetillwakeUp 5000
static TimerEvent_t sleep;
static TimerEvent_t wakeUp;
uint8_t lowpower=1;
#ifndef LoraWan_RGB
#define LoraWan_RGB 0
#endif

#define RF_FREQUENCY 868500000 // Hz
#define TX_OUTPUT_POWER 14 // dBm
#define LORA_BANDWIDTH 0 // [0: 125 kHz,
                          // 1: 250 kHz,
                          // 2: 500 kHz,
                          // 3: Reserved]
#define LORA_SPREADING_FACTOR 9 // [SF7..SF12]
#define LORA_CODINGRATE 1 // [1: 4/5,
                           // 2: 4/6,
                           // 3: 4/7,
                           // 4: 4/8]
#define LORA_PREAMBLE_LENGTH 8 // Same for Tx and Rx
#define LORA_SYMBOL_TIMEOUT 0 // Symbols
#define LORA_FIX_LENGTH_PAYLOAD_ON false
#define LORA_IQ_INVERSION_ON false
#define RX_TIMEOUT_VALUE 1000
#define BUFFER_SIZE 30 // Define the payload size here

char txPacket[BUFFER_SIZE];
static RadioEvents_t RadioEvents;
void OnTxDone( void );
void OnTxTimeout( void );

typedef enum {LOWPOWER,ReadVoltage,TX} States_t;
States_t state;

bool sleepMode = false;
int16_t rssi,rxSize;
uint16_t voltage;

void onSleep()
{
    Serial.printf("Going into lowpower mode, %d ms later wake up.\r\n",timetillwakeUp);
    lowpower=1; turnOffRGB(); //Radio.Sleep();
    //timetillwakeUp ms later wake up;
    TimerSetValue( &wakeUp, timetillwakeUp );
    TimerStart( &wakeUp );
}
void onWakeUp()
{
    Serial.printf("Woke up, %d ms later into lowpower mode.\r\n",timetillsleep);
    lowpower=0;
    //timetillsleep ms later into lowpower mode;
    TimerSetValue( &sleep, timetillsleep );
    TimerStart( &sleep );
}
```



```

void setup() {
    Serial.begin(9600);
    voltage = 0;
    rssi=0;
    RadioEvents.TxDone = OnTxDone;
    RadioEvents.TxTimeout = OnTxTimeout;
    Radio.Init( &RadioEvents );
    Radio.SetChannel( RF_FREQUENCY );
    Radio.SetTxConfig( MODEM_LORA, TX_OUTPUT_POWER, 0, LORA_BANDWIDTH,
                        LORA_SPREADING_FACTOR, LORA_CODINGRATE,
                        LORA_PREAMBLE_LENGTH, LORA_FIX_LENGTH_PAYLOAD_ON,
                        true, 0, 0, LORA_IQ_INVERSION_ON, 3000 );

    state=ReadVoltage;
    Radio.Sleep( ); // LoRa modem sleep mode
    TimerInit( &sleep, onSleep );
    TimerInit( &wakeUp, onWakeUp );
    onSleep();
}

void loop()
{
    if(lowpower) { lowPowerHandler(); }
    else
    {
        switch(state)
        {
            case TX:
            {
                sprintf(txPacket,"%s","ADC_battery (mV): ");
                sprintf(txPacket+strlen(txPacket), "%d", voltage);
                if(voltage<(uint16_t)3680)turnOnRGB(COLOR_SEND,0);
                else turnOnRGB(COLOR_RECEIVED,0);
                Serial.printf("\r\nsending packet \"%s\" , length %d\r\n",txPacket, strlen(txPacket));
                Radio.Send( (uint8_t *)txPacket, strlen(txPacket) );
                state=LOWPOWER;delay(100);
                break;
            }
            case LOWPOWER:
            {
                Serial.println("going to low power mode");
                delay(100);
                turnOffRGB();
                delay(100); //LowPower time
                state = ReadVoltage;
                break;
            }
            case ReadVoltage:
            {
                Serial.println("reading battery voltage");
                pinMode(VBAT_ADC_CTL,OUTPUT);
                digitalWrite(VBAT_ADC_CTL,LOW);
                voltage=analogRead(ADC);
                pinMode(VBAT_ADC_CTL, INPUT);
                state = TX;
                break;
            }
            default: break;
        }
        Radio.IrqProcess();
    }
}

void OnTxDone( void )
{
    Serial.println("TX done!");
    turnOffRGB();Radio.Sleep( );
}

void OnTxTimeout( void )
{
    Radio.Sleep( );
    Serial.println("TX Timeout.....");
    state=ReadVoltage;
    Serial.println(state);
}

```



Fig 2.2 Sending three LoRa packets




Fig 2.3 Sending three LoRa packets in **high_power** window (1000 ms).

Power/current consumption during the **high_power** (up to 200 mA) and **low_power** (19.93 µA) states.

To do

Analyze and test the above example.

```
Radio.IrqProcess();
Radio.Send( (uint8_t *)txPacket, strlen(txPacket) );
Radio.Sleep( );
```

2.3 LoRa receiver - receiving the packets with battery state

At the reception side we use the same headers as in the sender code to keep the identical parameters. The receiver is always active but it stays in sleep mode awaiting the new packet that is signaled by the associated interruption captured by `Radio.IrqProcess()`;

```
#include "LoRaWan_APP.h"
#include "Arduino.h"
#ifndef LoraWan_RGB
#define LoraWan_RGB 0
#endif

#define RF_FREQUENCY                868500000 // Hz
#define TX_OUTPUT_POWER             14        // dBm
#define LORA_BANDWIDTH              0        // [0: 125 kHz,
                                           // 1: 250 kHz,
                                           // 2: 500 kHz,
                                           // 3: Reserved]
#define LORA_SPREADING_FACTOR       9        // [SF7..SF12]
#define LORA_CODINGRATE             1        // [1: 4/5,
                                           // 2: 4/6,
                                           // 3: 4/7,
                                           // 4: 4/8]
#define LORA_PREAMBLE_LENGTH        8        // Same for Tx and Rx
#define LORA_SYMBOL_TIMEOUT         0        // Symbols
#define LORA_FIX_LENGTH_PAYLOAD_ON false
#define LORA_IQ_INVERSION_ON        false
#define RX_TIMEOUT_VALUE            1000
#define BUFFER_SIZE                 30 // Define the payload size here

char txpacket[BUFFER_SIZE];
char rxpacket[BUFFER_SIZE];

static RadioEvents_t RadioEvents;
int16_t txNumber;
int16_t rssi, rxSize;

void setup() {
    Serial.begin(9600);

    txNumber=0;
    rssi=0;

    RadioEvents.RxDone = OnRxDone;
    Radio.Init( &RadioEvents );
    Radio.SetChannel( RF_FREQUENCY );

    Radio.SetRxConfig( MODEM_LORA, LORA_BANDWIDTH, LORA_SPREADING_FACTOR,
                       LORA_CODINGRATE, 0, LORA_PREAMBLE_LENGTH,
                       LORA_SYMBOL_TIMEOUT, LORA_FIX_LENGTH_PAYLOAD_ON,
                       0, true, 0, 0, LORA_IQ_INVERSION_ON, true );

    turnOnRGB(COLOR_SEND,0); //change rgb color
    Serial.println("into RX mode");
}

void loop()
{
    Radio.Rx( 0 );
    delay(500);
    Radio.IrqProcess( );
}

void OnRxDone( uint8_t *payload, uint16_t size, int16_t rssi, int8_t snr )
{
    rssi=rssi;
    rxSize=size;
    memcpy(rxpacket, payload, size );
    rxpacket[size]='\0';
    turnOnRGB(COLOR_RECEIVED,0);
    Radio.Sleep( );
    Serial.printf("\r\nreceived packet \"%s\" with rssi %d , length %d\r\n",rxpacket,rssi,rxSize);
}

}
```

The IDE terminal output (fragment):

```
..
received packet "ADC_battery (mV) : 3886" with rssi -41 , length 22
received packet "ADC_battery (mV) : 3886" with rssi -40 , length 22
received packet "ADC_battery (mV) : 3886" with rssi -41 , length 22
received packet "ADC_battery (mV) : 3886" with rssi -39 , length 22
received packet "ADC_battery (mV) : 3886" with rssi -40 , length 22
received packet "ADC_battery (mV) : 3886" with rssi -41 , length 22
received packet "ADC_battery (mV) : 3886" with rssi -39 , length 22
received packet "ADC_battery (mV) : 3886" with rssi -40 , length 22
..
```




Fig 2.4 Reception of 2 LoRa packets (always active state)

To do

1. Analyze and test the above example.
2. Add an OLED display to show the received date (battery state)

2.4 Sending LoRa packets with external sensor data

In this section we are going to add an external sensor connected via **I2C** bus. This sensor (SHT21) will provide us with two data values: temperature and humidity. In this context you can use any other sensor to provide luminosity, CO2, atmospheric pressure, etc.




Fig 2.5 IoT CubeCell DevKit with several sensors.

The following **power consumption diagram** for the terminal node shows the current values required for different period and phases of the operational cycle.




Fig 2.6 Power consumption diagram (current for 3.3V) with low and high power periods and high-power operational phases.

SF	Chirps / Symbol	SNR	Airtime ^a	Bitrate
7	128	-7.5	56.5 ms	5469 bps
8	256	-10	103 ms	3125 bps
9	512	-12.5	185.3 ms	1758 bps
10	1024	-15	371 ms	977 bps
11	2048	-17.5	741 ms	537 bps
12	4096	-20	1318.9 ms	293 bps

^a 20 bytes per packet and Code Rate = 4/5.

Fig 2.7 Transmission – airtime during high power period for 20-byte payload and 4/5 coderate

Example of power consumption evaluation in µA for SF=8, CR=4/8, and 16 byte payload:

- airtime = $103 \cdot (16/20) \cdot (8/5)$ ms = $103 \cdot 128/100$ = 132 ms (less than 200 ms)
- transmission current for 14 dBm (25 mW radio) we need about 40 mA with 3.3V

Let us calculate the average current consumption for :

- 600 sec **low_power** period
- 2.2 sec **high_power** period with 3 phases

$$(600 \cdot 25 + 1 \cdot 10000 + 0.2 \cdot 60000 + 1 \cdot 10000) / 603 = 157,545605307 \Rightarrow 80 \mu A$$

For a battery with the capacity of 1000 mAh.

$$1000 \cdot 1000 / 80 = 6410,25 \Rightarrow 12500 \text{ hours or } 12500 / 750 \Rightarrow 16 \text{ months}$$

That is more than a year of operation.

In general the transmitting current depends on TX power set to :

- for 14 dBm (25 mW radio) we need about 40 mA (3.3V)
- for 17 dBm (50 mW radio) we need about 60 mA (3.3V)
- for 20 dBm (100 mW radio) we need about 100 mA (3.3V)

The following figure shows the transmission time as a function of payload size and the spreading factor value.




Fig 2.8 Time on Air for different spreading factors and payload size for 125 kHz band and 4/5 coding rate.

If we use different coding rate such as 4/8 we need to modify the obtained value with the 8/5 factor.




Fig 2.9 Time on Air for different payload size and coding rate ($CR=4/5$ to $CR=4/8$).

In the proposed example code we send 20-byte payload frame.

Note that our DevKit may be completed with a small solar providing max 100 mA (6V) cell such as:

As we need about 200 μ A of continuous power supply only 2% of the cell efficiency is enough to keep our terminal node running.



2.7 Terminal and the gateway codes

The terminal part of this code contains a number of timing parameters. The initial values for time to sleep and time to wake up. Consequently, the time to wake up value is modified with the received timeout value provide by the gateway. The time till sleep is fixed and is related to the timing of the sensor capture and the time necessary to wait for the ACK packet. ACK packet contains new time to wake up value (if not equal to 0).

- `uint32_t timetillsleep=5000;`
- `uint32_t timetillwakeup=12000;`

The following sequence of instructions determines the reception phase during the **high-power period** (time to sleep).

```
    long debut, del=2000;
    // the receiver must respond in this waiting period - reception window
..
    turnOnRGB(COLOR_SEND, 0);
    debut=millis(); delay(300); turnOffRGB(); Radio.Rx(0);
    while(millis() < (debut+del))
    {
        Radio.IrqProcess(); delay(100);
    }
```

2.7.1 The code of terminal node

```
#include "Arduino.h"
#define LoraWan_RGB 5
#include "LoRaWan_APP.h"
#include <Wire.h>
#include "SHT21.h"
SHT21 SHT21;
#define RF_FREQUENCY 868500000 // Hz
#define TX_OUTPUT_POWER 14 // dBm - 25mW, 17-50mW, 20 - 100mW
#define LORA_BANDWIDTH 0 // [0: 125 kHz,
                          // 1: 250 kHz,
                          // 2: 500 kHz,
                          // 3: Reserved]
// 3: Reserved]
#define LORA_SPREADING_FACTOR 9 // [SF7..SF12]
#define LORA_CODINGRATE 1 // [1: 4/5,
                          // 2: 4/6,
                          // 3: 4/7,
                          // 4: 4/8]
#define LORA_PREAMBLE_LENGTH 8 // Same for Tx and Rx
#define LORA_SYMBOL_TIMEOUT 0 // Symbols
#define LORA_FIX_LENGTH_PAYLOAD_ON false
#define LORA_IQ_INVERSION_ON false
#define RX_TIMEOUT_VALUE 1000
#define BUFFER_SIZE 30 // Define the payload size here

char txPacket[BUFFER_SIZE];
static RadioEvents_t RadioEvents;
void OnTxDone( void );
void OnTxTimeout( void );
uint32_t timetillsleep=5000;
uint32_t timetillwakeup=12000;
static TimerEvent_t sleepGo;
static TimerEvent_t wakeUp;
uint8_t lowpower=1, highpower=0;
int16_t txNumber;
bool sleepMode = false;
int16_t rssi, rxSize;
uint16_t voltage;
float t, h;
int dt, dh;
char buff[32];

union
{
    uint8_t frame[16];
    float sensor[4];
} sdp; // send data packet
```

```

void getSHT21()
{
    pinMode(Vext, OUTPUT);
    digitalWrite(Vext, LOW);delay(200);
    Wire.begin(29,28);
    SHT21.begin();
    sdp.sensor[0]=SHT21.getTemperature();
    sdp.sensor[1]=SHT21.getHumidity();
    delay(100);
    Wire.end();
    Serial.println(sdp.sensor[0]);Serial.println(sdp.sensor[1]);
    digitalWrite(Vext, HIGH);
}

void onSleep()
{
    Serial.printf("Going into lowpower mode, %d ms later wake up.\r\n",timetillwakeup);
    lowpower=1;highpower=0;
    //timetillwakeup ms later wake up;
    TimerSetValue( &wakeUp, timetillwakeup );
    TimerStart( &wakeUp );
}

void onWakeUp()
{
    Serial.printf("Woke up, %d ms later into lowpower mode.\r\n",timetillsleep);
    lowpower=0;highpower=1; //turnOnRGB(0,0);
    //timetillsleep ms later into lowpower mode;
    TimerSetValue( &sleepGo, timetillsleep );
    TimerStart( &sleepGo );
}

void OnTxDone( void )
{
    Serial.println("TX done!");Radio.Sleep( );
}

void OnTxTimeout( void )
{
    Radio.Sleep( );Serial.println("TX Timeout.....");
}

void setup() {
    Serial.begin(9600);
    pinMode(Vext,OUTPUT);
    digitalWrite(Vext,LOW); //SET POWER
    voltage = 0;
    rssi=0;
    RadioEvents.TxDone = OnTxDone;
    RadioEvents.TxTimeout = OnTxTimeout;
    Radio.Init( &RadioEvents );
    Radio.SetChannel( RF_FREQUENCY );
    Radio.SetTxConfig( MODEM_LORA, TX_OUTPUT_POWER, 0, LORA_BANDWIDTH,
                      LORA_SPREADING_FACTOR, LORA_CODINGRATE,
                      LORA_PREAMBLE_LENGTH, LORA_FIX_LENGTH_PAYLOAD_ON,
                      true, 0, 0, LORA_IQ_INVERSION_ON, 3000 );

    //state=ReadVoltage;
    Radio.Sleep( );
    TimerInit( &sleepGo, onSleep ); // activate on event
    TimerInit( &wakeUp, onWakeUp );
    onSleep();
}

long debut,del=2000; // the receiver must respond in this waiting period

void loop() {
    if(lowpower){
        lowPowerHandler();
    }
    if(highpower)
    {
        Serial.println("in highpower");
        getSHT21();
        pinMode(VBAT_ADC_CTL,OUTPUT);
        digitalWrite(VBAT_ADC_CTL,LOW);
        voltage=analogRead(ADC)+400;
    }
}

```



```

pinMode(VBAT_ADC_CTL, INPUT);
sprintf(txPacket, "V: %d, T: %d, H: %d", voltage, (int)sdp.sensor[0], (int)sdp.sensor[1]);
Serial.printf("sending packet [%s], length: %d\n", txPacket, strlen(txPacket));
delay(100);
Radio.Send( (uint8_t *)txPacket, strlen(txPacket) );
highpower=0;lowpower=1;
}
Radio.IrqProcess( );
}

```



Fig 2.10 (a) (b) low_power and high_power sensing periods




Fig 2.10 (c) high_power transmission period

```

..
in highpower
21.23
35.37
sending packet [V:4028,T:21,H:35], length: 16
TX Going into lowpower mode, 12000 ms later w¶Woke up, 5000 ms later into lowpower mode.
in highpower
21.20
35.44
sending packet [V:4046,T:21,H:35], length: 16
TX Going into lowpower mode, 12000 ms later w¶Woke up, 5000 ms later into lowpower mode.
in highpower
21.17
35.37
sending packet [V:4046,T:21,H:35], length: 16
TX Going into lowpower mode, 12000 ms later w¶Woke up, 5000 ms later into lowpower mode.
in highpower
21.18
37.14
sending packet [V:4046,T:21,H:37], length: 16
TX Going into lowpower mode, 12000 ms later w¶Woke up, 5000 ms later into lowpower mode.
in highpower
..

```

To do:

- Test the above codes
- Modify the radio parameters such as **spreading factor**, **signal bandwidth** and the **transmission power** and observe the current consumption

2.8 LoRa-WiFi Gateways for CubeCell terminals

In this section we introduce some simple LoRa to WiFi gateways that allow us to receive the packets from low power terminals such as CubeCell DevKit and to relay the received data to IoT servers (MQTT, ThingSpeak,..)

The first example is developed on Lolin D32 based board with LoRa (RFM95-SX1276) modem (868MHz).



Fig 2.11 LoRa receiver on Lolin D32 based board

2.8.1 LoRa receiver (+display) for CubeCell terminals

The following receiver code corresponds to the presented above low power sender example. In the header we have added `ThingSpeak.h` and `WiFi.h` libraries. They will be effectively used in the following examples.

```
#include <SPI.h>
#include <LoRa.h>
#include <WiFi.h>
#include <Wire.h>
#include "SSD1306Wire.h"
SSD1306Wire display(0x3c, 12, 14);
#include "ThingSpeak.h"
char* ssid = "PhoneAP";
char* pass = "smartcomputerlab";
WiFiClient client;

unsigned long myChannelNumber = 114; // Thinspeak channel
const char *myWriteAPIKey="E8KYBCRD2Z59LVWJ" ; // write code

#define SCK      18 // GPIO18 -- SX127x's SCK
#define MISO     19 // GPIO19 -- SX127x's MISO
#define MOSI     23 // GPIO23 -- SX127x's MOSI
#define SS       5  // GPIO05 -- SX127x's CS
#define RST      15 // GPIO15 -- SX127x's RESET
#define DIO      26 // GPIO26 -- SX127x's IRQ(Interrupt Request)
#define freq     8685E5
#define sf       9
#define sb       125E3

union
{
  uint8_t frame[64]; // trames avec octets
  char mess[64]; // text message
  float sensor[8]; // 4 valeurs en virgule flottante
} rdp;

void disp(char *d1,char *d2,char *d3)
{
  display.init();
  //display.flipScreenVertically();
  display.setTextAlignment(TEXT_ALIGN_LEFT);
  display.setFont(ArialMT_Plain_16); // ArialMT_Plain_10
  display.drawString(0, 0, d1);
  display.drawString(0, 18, d2);
  display.drawString(0, 36, d3);
  display.setFont(ArialMT_Plain_10);
```

```

    display.drawString(20, 52, "SmartComputerLab");
    display.display();
}

void setup() {
    Serial.begin(9600);
    SPI.begin(SCK,MISO,MOSI,SS);
    LoRa.setPins(SS,RST,DIO);
    Serial.println();delay(100);Serial.println();
    if (!LoRa.begin(freq)) {
        Serial.println("Starting LoRa failed!");
        while (1);
    }
    Serial.println("Starting LoRa OK!");
    LoRa.setSpreadingFactor(sf);
    LoRa.setSignalBandwidth(sb);
}

void loop()
{
    int packetLen, rssi;
    packetLen=LoRa.parsePacket();
    if(packetLen)
    {
        int i=0; char tbuff[32];
        while (LoRa.available()){rdp.frame[i]=LoRa.read();i++;}
        rssi=LoRa.packetRssi(); // force du signal en réception en dB
        Serial.printf("Received packet:%s\n",rdp.mess+12);
        Serial.printf("RSSI=%d\n",rssi); sprintf(tbuff,"RSSI= %d",rssi);
        disp(rdp.mess+12,tbuff," ");delay(1000);
    }
}

```

2.8.2 LoRa-WiFi gateway to ThingSpeak (+display) for CubeCell terminals

```

#include <SPI.h>
#include <LoRa.h>
#include <WiFi.h>
#include <Wire.h>
#include "SSD1306Wire.h"
SSD1306Wire display(0x3c, 12, 14);
#include "ThingSpeak.h"
//char* ssid      = "PhoneAP";
//char* pass      = "smartcomputerlab";
char* ssid      = "Livebox-08B0";
char* pass      = "G79ji6dtEptVTPWmZP";
WiFiClient client;

unsigned long myChannelNumber = 1697980; // Thinspeak channel - tag: smartiotlabs-2
const char *myWriteAPIKey="4K897XNNHTW7I4NO" ; // write code

#define SCK      18 // GPIO18 -- SX127x's SCK
#define MISO     19 // GPIO19 -- SX127x's MISO
#define MOSI     23 // GPIO23 -- SX127x's MOSI
#define SS       5  // GPIO05 -- SX127x's CS
#define RST      15 // GPIO15 -- SX127x's RESET
#define DIO      26 // GPIO26 -- SX127x's IRQ(Interrupt Request)
#define freq     8685E5
#define sf       9
#define sb       125E3

union
{
    uint8_t frame[64]; // trames avec octets
    char mess[64]; // text message
    float sensor[8]; // 4 valeurs en virgule flottante
} rdp;

void disp(char *d1,char *d2,char *d3)
{
    display.init();
    //display.flipScreenVertically();
    display.setTextAlignment(TEXT_ALIGN_LEFT);
    display.setFont(ArialMT_Plain_16); // ArialMT_Plain_10

```

```

    display.drawString(0, 0, d1);
    display.drawString(0, 18, d2);
    display.drawString(0, 36, d3);
    display.setFont(ArialMT_Plain_10);
    display.drawString(20, 52, "SmartComputerLab");
    display.display();
}

void setup() {
    Serial.begin(9600);
    WiFi.disconnect(true); WiFi.mode(WIFI_STA);
    WiFi.begin(ssid, pass);
    while (WiFi.status() != WL_CONNECTED) {
        delay(500); Serial.print(".");
    }
    IPAddress ip = WiFi.localIP();
    Serial.print("IP Address: "); Serial.println(ip);
    ThingSpeak.begin(client); delay(1000);
    SPI.begin(SCK, MISO, MOSI, SS);
    LoRa.setPins(SS, RST, DI0);
    Serial.println(); delay(100); Serial.println();
    if (!LoRa.begin(freq)) {
        Serial.println("Starting LoRa failed!");
        while (1);
    }
    Serial.println("Starting LoRa OK!");
    LoRa.setSpreadingFactor(sf);
    LoRa.setSignalBandwidth(sb);
}

uint32_t mindel=15000; // 15 seconds
void loop()
{
    int packetLen, rssi;
    packetLen=LoRa.parsePacket();
    if(packetLen)
    {
        int i=0; char tbuff[32];
        int volt=0;
        while (LoRa.available()){rdp.frame[i]=LoRa.read();i++;}
        rssi=LoRa.packetRssi(); // force du signal en réception en dB
        Serial.printf("Received packet:%s\n", rdp.mess+12);
        Serial.printf("RSSI=%d\n", rssi); sprintf(tbuff, "RSSI= %d", rssi);
        disp(rdp.mess+12, tbuff, " ");
        volt= atoi(rdp.mess+17);
        ThingSpeak.setField(1, rssi);
        ThingSpeak.setField(2, volt);
        // ThingSpeak.setField(3, rp.data[2]);
        // ThingSpeak.setField(4, rp.data[3]);
        while (WiFi.status() != WL_CONNECTED) { delay(500); }
        ThingSpeak.writeFields(myChannelNumber, myWriteAPIKey);
        delay(mindel);
    }
}

```



[+ Add Visualizations](#)[+ Add Widgets](#)[Export recent data](#)[MATLAB Analysis](#)[MATLAB Visualization](#)

Channel Stats

Created: [9 months ago](#)

Last entry: [less than a minute ago](#)

Entries: 58



Assignment

1. Complete the code for CubeCell boards adding some sensors.
2. Use the union structure for the preparation of data packets

```
union
{
    uint8_t frame[64]; // trames avec octets
    char    mess[64];  // text message
    float   sensor[8]; // 4 valeurs en virgule flottante
} rdp;
```

3. Rewrite the code for the LoRa-WiFi gateway on your Lolin D32 based board in order to receive the sensor data sent by the terminal.
4. Use the PPK2 power profiler to illustrate the power consumption of the terminal with different spreading factors (to be adapted on the gateway)